```
> restart:
```

Università delgi studi di Roma La Sapienza

Formule di riferimento per ripetere le lezioni
di Meccanica applicata alle macchine (meccanici)

A.A. 2009 - 2010

ANALISI CINEMATICA MEDIANTE IL METODO DELLE EQUAZIONI DI VINCOLO
(EQUAZIONI DI CHIUSURA ED ANGOLI DELLE ASTE)

Si prende 1 coordinata angolare per ogni corpo mobile (nel piano)
theta_i è l'angolo di posizione del riferimento piano solidale al corpo i-esimo rispetto al riferimento piano fissato al telaio

nel quadrilatero si ha [q] := [theta_3,theta_4,theta_2]T che si partiziona nei due vettori
[u] := [theta_3,theta_4] delle variabili dipendenti e
[v] :=[theta_2] delle variabili indipendenti (corrispondono ai moventi)

In questo esempio si ha:
n = 3 coordinate lagrangiane
p = 2 coordinate di vincolo dedotte delle equazioni di chiusura del sistema
F = n - p = 1 gradi di libertà, ovvero dimensioni del vettore [v] delle coordinate lagrangiane moventi.

```
> with(LinearAlgebra):
```

DEFINIZIONE DEI VINCOLI
Definizione delle equazioni di DI CHIUSURA

```
> psi := Vector(2):
  vincoloX := l[2]*cos(theta2)+ l[3]*cos(theta3)+l[4]*cos(theta4)-l[1];
  vincoloY := l[2]*sin(theta2)+ l[3]*sin(theta3)+l[4]*sin(theta4);
  psi[1] := vincoloX;
  psi[2] := vincoloY;
```

$$vincoloX := l_2 \cos(\theta 2) + l_3 \cos(\theta 3) + l_4 \cos(\theta 4) - l_1$$

$$vincoloY := l_2 \sin(\theta 2) + l_3 \sin(\theta 3) + l_4 \sin(\theta 4)$$

$$\psi_1 := l_2 \cos(\theta 2) + l_3 \cos(\theta 3) + l_4 \cos(\theta 4) - l_1$$

$$\psi_2 := l_2 \sin(\theta 2) + l_3 \sin(\theta 3) + l_4 \sin(\theta 4)$$

Vettore dei vincoli Psi(q) = 0

```
> psi;
```

$$\begin{bmatrix} l_2 \cos(\theta 2) + l_3 \cos(\theta 3) + l_4 \cos(\theta 4) - l_1 \\ l_2 \sin(\theta 2) + l_3 \sin(\theta 3) + l_4 \sin(\theta 4) \end{bmatrix}$$

Le coordinate lagrangiane sono in questo caso le 3 coordinate assolute (tre angoli)
essendo 2 le equazioni di vincolo abbiamo un solo parametro libero (corrispondente al grado di libertà del sistema)

INCOGNITE E SET di EQUAZIONI

```
> set_lagrangiane := {theta3,theta4,theta2};
  set_incognite := {theta3,theta4};
  set_espressioni := {}:
  for i from 1 to 2 do set_espressioni := set_espressioni union
  {psi[i]} od:
  set_espressioni;
  qq := Vector([theta3,theta4,theta2 ]);
  uu := SubVector(qq,[1..2]);
  vv := SubVector(qq,[3]);
```

$$set\_lagrangiane := \{\theta 2, \theta 3, \theta 4\}$$

$$set\_incognite := \{\theta 3, \theta 4\}$$

$$\{l_2 \sin(\theta 2) + l_3 \sin(\theta 3) + l_4 \sin(\theta 4), l_2 \cos(\theta 2) + l_3 \cos(\theta 3) + l_4 \cos(\theta 4) - l_1\}$$

$$qq := \begin{bmatrix} \theta 3 \\ \theta 4 \\ \theta 2 \end{bmatrix}$$

$$uu := \begin{bmatrix} \theta 3 \\ \theta 4 \end{bmatrix}$$

$$vv := \begin{bmatrix} \theta 2 \end{bmatrix}$$

Espressione dello jacobiano del vettore delle equazioni di vincolo e suoi minori di interesse

```
> psi_q := Matrix(2,3,linalg[ jacobian ](psi,[theta3,theta4,
  theta2]));
  psi_u := Matrix(2,2,SubMatrix(psi_q,1..2,1..2));
  psi_v := Matrix(2,1,SubMatrix(psi_q,1..2,3..3));
```

$$psi\_q := \begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) & -l_2 \sin(\theta 2) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) & l_2 \cos(\theta 2) \end{bmatrix}$$

$$psi\_u := \begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) \end{bmatrix}$$

$$psi\_v := \begin{bmatrix} -l_2 \sin(\theta 2) \\ l_2 \cos(\theta 2) \end{bmatrix}$$

Determinante del minore dello jacobiano psi_y

```
> Det_psi_u := simplify(Determinant(psi_u));
```

$$Det\_psi\_u := -l_3\, l_4\, (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))$$

Matrice inversa del minore psi_y

```
> Inv_psi_u := MatrixInverse(psi_u);
```

$$Inv\_psi\_u := \left[\left[-\frac{\cos(\theta 4)}{l_3\, (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}, -\frac{\sin(\theta 4)}{l_3\, (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}\right], \left[\frac{\cos(\theta 3)}{l_4\, (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}, \frac{\sin(\theta 3)}{l_4\, (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}\right]\right]$$

VALORI NOTI - semplificazione delle espressioni

```
> v_punto := Vector([omega2]);
```

$$v\_punto := \begin{bmatrix} \omega 2 \end{bmatrix}$$

Espressione del determinante della psi_y e della inversa (abbreviato)

```
> D_psi_u := Det_psi_u;
  I_psi_u := Inv_psi_u;
  ps_v := psi_v;
  ps := psi;
  ps_q := psi_q;
  ps_u := psi_u;
```

$$D\_psi\_u := -l_3\, l_4\, (\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3))$$

$$I\_psi\_u := \left[\left[-\frac{\cos(\theta 4)}{l_3\,(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3))}, -\frac{\sin(\theta 4)}{l_3\,(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3))}\right], \left[\frac{\cos(\theta 3)}{l_4\,(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3))}, \frac{\sin(\theta 3)}{l_4\,(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3))}\right]\right]$$

$$ps\_v := \begin{bmatrix} -l_2 \sin(\theta 2) \\ l_2 \cos(\theta 2) \end{bmatrix}$$

$$ps := \begin{bmatrix} l_2 \cos(\theta 2) + l_3 \cos(\theta 3) + l_4 \cos(\theta 4) - l_1 \\ l_2 \sin(\theta 2) + l_3 \sin(\theta 3) + l_4 \sin(\theta 4) \end{bmatrix}$$

$$ps\_q := \begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) & -l_2 \sin(\theta 2) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) & l_2 \cos(\theta 2) \end{bmatrix}$$

$$ps\_u := \begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) \end{bmatrix}$$

Matrice (in questo caso vettore) dei Rapporti di velocità e vettore u_punto

```
> Rap := MatrixMatrixMultiply(I_psi_u,ps_v):
  for ii from 1 to 2 do for jj from 1 to 1 do Rap[ii,jj] :=
  combine(simplify(Rap[ii,jj]),trig) od: od:
  Rap;
  u_punto := - MatrixVectorMultiply(Rap,v_punto);
```

$$\begin{bmatrix} \dfrac{l_2 \sin(-\theta 4 + \theta 2)}{l_3 \sin(\theta 3 - \theta 4)} \\ -\dfrac{l_2 \sin(-\theta 3 + \theta 2)}{l_4 \sin(\theta 3 - \theta 4)} \end{bmatrix}$$

$$u\_punto := \begin{bmatrix} -\dfrac{l_2 \sin(-\theta 4 + \theta 2)\ \omega 2}{l_3 \sin(\theta 3 - \theta 4)} \\ \dfrac{l_2 \sin(-\theta 3 + \theta 2)\ \omega 2}{l_4 \sin(\theta 3 - \theta 4)} \end{bmatrix}$$

CALCOLO DELLE VELOCITA' e delle ACCELERAZIONI

```
> q_punto := Vector([u_punto[1],u_punto[2],omega2]);
```

$$q\_punto := \begin{bmatrix} -\dfrac{l_2 \sin(-\theta 4 + \theta 2)\ \omega 2}{l_3 \sin(\theta 3 - \theta 4)} \\ \dfrac{l_2 \sin(-\theta 3 + \theta 2)\ \omega 2}{l_4 \sin(\theta 3 - \theta 4)} \\ \omega 2 \end{bmatrix}$$

matrici psi_u e psi_v in forma di funzioni del tempo

```
> d_ps_u := subs(theta3=t3(t),theta4=t4(t),ps_u);
  d_ps_v := subs(theta2=t2(t),ps_v);
```

$$d\_ps\_u := \begin{bmatrix} -l_3 \sin(t3(t)) & -l_4 \sin(t4(t)) \\ l_3 \cos(t3(t)) & l_4 \cos(t4(t)) \end{bmatrix}$$

$$d\_ps\_v := \begin{bmatrix} -l_2 \sin(t2(t)) \\ l_2 \cos(t2(t)) \end{bmatrix}$$

matrice psi_u_punto, ovvero derivata temporale degoli elementi della psi_u

```
> d_ps_u_pu := Matrix(2,2):
  for ii from 1 to 2 do for jj from 1 to 2 do
  appoggio := diff(d_ps_u[ii,jj],t);
  d_ps_u_pu[ii,jj] := subs(diff(t4(t),t)=omega4,diff(t3(t),t)=
  omega3,t3(t)=theta3,t4(t)=theta4,t2(t)=theta2,appoggio)
  od; od;
  d_ps_u_pu;
```

$$\begin{bmatrix} -l_3 \cos(\theta 3)\ \omega 3 & -l_4 \cos(\theta 4)\ \omega 4 \\ -l_3 \sin(\theta 3)\ \omega 3 & -l_4 \sin(\theta 4)\ \omega 4 \end{bmatrix}$$

matrice psi_u = ([psi_u] * {u_pu})_u
 ottenuta col procedimento ([psi_u]{u_punto})_u {u_punto}

Nota che si ottiene lo stesso risultato

```
> psi_u;
  u_punt := Vector([omega3,omega4]);
  Jac01 := MatrixVectorMultiply(psi_u,u_punt);
  Matrix(2,2,linalg[ jacobian ](Jac01,[theta3,theta4]));
```

$$\begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) \end{bmatrix}$$

$$u\_punt := \begin{bmatrix} \omega 3 \\ \omega 4 \end{bmatrix}$$

$$Jac01 := \begin{bmatrix} -l_3 \sin(\theta 3)\ \omega 3 - l_4 \sin(\theta 4)\ \omega 4 \\ l_3 \cos(\theta 3)\ \omega 3 + l_4 \cos(\theta 4)\ \omega 4 \end{bmatrix}$$

$$\begin{bmatrix} -l_3 \cos(\theta 3)\ \omega 3 & -l_4 \cos(\theta 4)\ \omega 4 \\ -l_3 \sin(\theta 3)\ \omega 3 & -l_4 \sin(\theta 4)\ \omega 4 \end{bmatrix} \tag{1}$$

matrice psi_v_punto

```
> d_ps_v_pu := Matrix(2,1):
  for ii from 1 to 2 do
  appoggio2 := diff(d_ps_v[ii,1],t):
  d_ps_v_pu[ii,1] :=  subs(diff(t2(t),t)=omega2,diff(t3(t),t)=
  omega3,t3(t)=theta3,t4(t)=theta4,t2(t)=theta2,appoggio2)
  od:
  d_ps_v_pu;
```

$$\begin{bmatrix} -l_2 \cos(\theta 2)\ \omega 2 \\ -l_2 \sin(\theta 2)\ \omega 2 \end{bmatrix}$$

Matrice psi_v

```
> d_ps_v;
  d_ps_vv := subs(t2(t)=theta2,d_ps_v);
```

$$\begin{bmatrix} -l_2 \sin(t2(t)) \\ l_2 \cos(t2(t)) \end{bmatrix}$$

$$d\_ps\_vv := \begin{bmatrix} -l_2 \sin(\theta 2) \\ l_2 \cos(\theta 2) \end{bmatrix}$$

vettori u_punto e v_punto

```
> u_punto;
  v_punto;
```

$$\begin{bmatrix} -\dfrac{l_2 \sin(-\theta 4 + \theta 2)\ \omega 2}{l_3 \sin(\theta 3 - \theta 4)} \\ \dfrac{l_2 \sin(-\theta 3 + \theta 2)\ \omega 2}{l_4 \sin(\theta 3 - \theta 4)} \end{bmatrix}$$

$$\left[\ \alpha 2\ \right]$$

CALCOLO DELLE ACCELERAZIONI

VETTORE DELLE ACCELERAZIONI INDIPENDENTI

```
> v_2_pu := Vector([alpha2]);
```

$$v\_2\_pu := \left[\ \alpha 2\ \right] \qquad (2)$$

VETTORE H1 = [psi_v] * {v_2_punti}

```
> H1 := MatrixVectorMultiply(d_ps_vv,v_2_pu);
```

$$H1 := \begin{bmatrix} -l_2 \sin(\theta 2)\ \alpha 2 \\ l_2 \cos(\theta 2)\ \alpha 2 \end{bmatrix} \qquad (3)$$

VETTORE H2 = [psi_u_punto] * {u_punto}

```
> H2 := MatrixVectorMultiply(d_ps_u_pu,u_punto);
```

$$H2 := \begin{bmatrix} \dfrac{\cos(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4 + \theta 2)\ \omega 2}{\sin(\theta 3 - \theta 4)} - \dfrac{\cos(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3 + \theta 2)\ \omega 2}{\sin(\theta 3 - \theta 4)} \\ \dfrac{\sin(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4 + \theta 2)\ \omega 2}{\sin(\theta 3 - \theta 4)} - \dfrac{\sin(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3 + \theta 2)\ \omega 2}{\sin(\theta 3 - \theta 4)} \end{bmatrix} \qquad (4)$$

VETTORE   H3 = [psi_u_punto] * {u_punto}

```
> H3 := MatrixVectorMultiply(d_ps_v_pu,v_punto);
```

$$H3 := \begin{bmatrix} -l_2 \cos(\theta 2)\ \omega 2^2 \\ -l_2 \sin(\theta 2)\ \omega 2^2 \end{bmatrix} \qquad (5)$$

VETTORE H0 = H1 + H2 + H3

```
> H0 := H1 + H2 + H3;
```

$$H0 := \left[\left[-l_2 \sin(\theta 2)\ \alpha 2 + \frac{\cos(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4 + \theta 2)\ \omega 2}{\sin(\theta 3 - \theta 4)} \right.\right. \qquad (6)$$

$$-\frac{\cos(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)} - l_2 \cos(\theta 2)\ \omega 2^2\Bigg],$$

$$\Bigg[l_2 \cos(\theta 2)\ \alpha 2 + \frac{\sin(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)} - \frac{\sin(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)}$$

$$- l_2 \sin(\theta 2)\ \omega 2^2\Bigg]\Bigg]$$

VETTORE {u_2_punti} = - [Inversa_psi_u] * {H0}

```
> u_2_pu := - MatrixVectorMultiply(I_psi_u,H0);
```

$$u\_2\_pu := \Bigg[\Bigg[\frac{1}{l_3\left(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3)\right)}\Bigg(\cos(\theta 4)\Bigg(-l_2 \sin(\theta 2)\ \alpha 2 \tag{7}$$

$$+ \frac{\cos(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)} - \frac{\cos(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)}$$

$$- l_2 \cos(\theta 2)\ \omega 2^2\Bigg)\Bigg)$$

$$+ \frac{1}{l_3\left(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3)\right)}\Bigg(\sin(\theta 4)\Bigg(l_2 \cos(\theta 2)\ \alpha 2$$

$$+ \frac{\sin(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)} - \frac{\sin(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)}$$

$$- l_2 \sin(\theta 2)\ \omega 2^2\Bigg)\Bigg)\Bigg],$$

$$\Bigg[-\frac{1}{l_4\left(\sin(\theta 3)\cos(\theta 4) - \sin(\theta 4)\cos(\theta 3)\right)}\Bigg(\cos(\theta 3)\Bigg(-l_2 \sin(\theta 2)\ \alpha 2$$

$$+ \frac{\cos(\theta 3)\ \omega 3\ l_2 \sin(-\theta 4+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)} - \frac{\cos(\theta 4)\ \omega 4\ l_2 \sin(-\theta 3+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)}$$

$$- l_2 \cos(\theta 2)\ \omega 2^2\Bigg)\Bigg)$$

$$-\frac{1}{l_4\left(\sin(\theta 3)\cos(\theta 4)-\sin(\theta 4)\cos(\theta 3)\right)}\left(\sin(\theta 3)\left(l_2\cos(\theta 2)\ \alpha 2\right.\right.$$

$$+\frac{\sin(\theta 3)\ \omega 3\ l_2\sin(-\theta 4+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)}-\frac{\sin(\theta 4)\ \omega 4\ l_2\sin(-\theta 3+\theta 2)\ \omega 2}{\sin(\theta 3-\theta 4)}$$

$$\left.\left.-l_2\sin(\theta 2)\ \omega 2^2\right)\right)\Big]\Big]$$

Elementi della u_2_punti semplificati (primo e secondo)

```
> u_2_pu[1] := combine(simplify(u_2_pu[1]),trig);
  u_2_pu[2] := combine(simplify(u_2_pu[2]),trig);
```

$$u\_2\_pu_1 := \frac{1}{-l_3+l_3\cos(2\ \theta 3-2\ \theta 4)}\left(2\ l_2\ \omega 4\sin(-\theta 3+\theta 2)\ \omega 2+l_2\ \alpha 2\cos(-\theta 3+\theta 2)\right.$$

$$-l_2\ \alpha 2\cos(\theta 3-2\ \theta 4+\theta 2)-l_2\ \omega 3\ \omega 2\sin(\theta 3-2\ \theta 4+\theta 2)-l_2\ \omega 3\ \omega 2\sin(-\theta 3+\theta 2)$$

$$\left.+l_2\ \omega 2^2\sin(\theta 3-2\ \theta 4+\theta 2)-l_2\ \omega 2^2\sin(-\theta 3+\theta 2)\right)$$

$$u\_2\_pu_2 := \frac{1}{-l_4+l_4\cos(2\ \theta 3-2\ \theta 4)}\left(-l_2\ \omega 4\ \omega 2\sin(-\theta 4+\theta 2)-l_2\ \omega 4\ \omega 2\sin(-2\ \theta 3+\theta 4\right.$$

$$+\theta 2)+2\ l_2\ \omega 3\sin(-\theta 4+\theta 2)\ \omega 2-l_2\ \alpha 2\cos(-2\ \theta 3+\theta 4+\theta 2)+l_2\ \alpha 2\cos(-\theta 4$$

$$\left.+\theta 2)-l_2\ \omega 2^2\sin(-\theta 4+\theta 2)+l_2\ \omega 2^2\sin(-2\ \theta 3+\theta 4+\theta 2)\right)$$

METODO DELLO JACOBIANO DEL vettore { [psi_u] * {u_punto} }

Riepilogo di matrici e vettori

```
> psi_u;
  q_punt := Vector([omega3,omega4,omega2]);
  u_punt := Vector([omega3,omega4]);
  v_punt := Vector([omega2]);
  ps_q;
  ps_u;
  ps_v;
```

$$\begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) \end{bmatrix}$$

$$q\_punt := \begin{bmatrix} \omega 3 \\ \omega 4 \\ \omega 2 \end{bmatrix}$$

$$u\_punt := \begin{bmatrix} \omega 3 \\ \omega 4 \end{bmatrix}$$

$$v\_punt := \begin{bmatrix} \omega 2 \end{bmatrix}$$

$$\begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) & -l_2 \sin(\theta 2) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) & l_2 \cos(\theta 2) \end{bmatrix}$$

$$\begin{bmatrix} -l_3 \sin(\theta 3) & -l_4 \sin(\theta 4) \\ l_3 \cos(\theta 3) & l_4 \cos(\theta 4) \end{bmatrix}$$

$$\begin{bmatrix} -l_2 \sin(\theta 2) \\ l_2 \cos(\theta 2) \end{bmatrix}$$

Hu = psi_u * u_punto
Hv = psi_v * v_punto

```
> Hq := MatrixVectorMultiply(ps_q,q_punt);
  Hu := MatrixVectorMultiply(ps_u,u_punt);
  Hv := MatrixVectorMultiply(ps_v,v_punt);
```

$$Hq := \begin{bmatrix} -l_3 \sin(\theta 3)\ \omega 3 - l_4 \sin(\theta 4)\ \omega 4 - l_2 \sin(\theta 2)\ \omega 2 \\ l_3 \cos(\theta 3)\ \omega 3 + l_4 \cos(\theta 4)\ \omega 4 + l_2 \cos(\theta 2)\ \omega 2 \end{bmatrix}$$

$$Hu := \begin{bmatrix} -l_3 \sin(\theta 3)\ \omega 3 - l_4 \sin(\theta 4)\ \omega 4 \\ l_3 \cos(\theta 3)\ \omega 3 + l_4 \cos(\theta 4)\ \omega 4 \end{bmatrix}$$

$$Hv := \begin{bmatrix} -l_2 \sin(\theta 2)\ \omega 2 \\ l_2 \cos(\theta 2)\ \omega 2 \end{bmatrix}$$

calcolo della jacobiano della Hu

```
> Hq_q :=  Matrix(2,3,linalg[ jacobian ](Hq,[theta3,theta4,theta2]
  )):
  Hu_u :=  Matrix(2,2,linalg[ jacobian ](Hu,[theta3,theta4]));
  for ii from 1 to 2 do for jj from 1 to 2 do Hu_u[ii,jj] :=
  combine(simplify(Hu_u[ii,jj]),trig) od: od:
  Hu_u;
```

$$Hu\_u := \begin{bmatrix} -l_3 \cos(\theta 3)\ \omega 3 & -l_4 \cos(\theta 4)\ \omega 4 \\ -l_3 \sin(\theta 3)\ \omega 3 & -l_4 \sin(\theta 4)\ \omega 4 \end{bmatrix}$$

$$\begin{bmatrix} -l_3 \cos(\theta 3)\ \omega 3 & -l_4 \cos(\theta 4)\ \omega 4 \\ -l_3 \sin(\theta 3)\ \omega 3 & -l_4 \sin(\theta 4)\ \omega 4 \end{bmatrix}$$

calcolo dello jacobiano della Hv

```
> Hv_v :=  Matrix(2,1,linalg[ jacobian ](Hv,[theta2]));
```

$$Hv\_v := \begin{bmatrix} -l_2 \cos(\theta 2)\ \omega 2 \\ -l_2 \sin(\theta 2)\ \omega 2 \end{bmatrix}$$

calcolo del vettore gamma0 inteso come gamma1+gamma2 ove
gamma1 = Hu_u * u_punto
gamma2 = Hv_v * v_punto

```
> gamma1 := MatrixVectorMultiply(Hu_u,u_punt);
  gamma2 := MatrixVectorMultiply(Hv_v,v_punt);
  gamma0 :=  gamma1 + gamma2  ;
```

$$\gamma 1 := \begin{bmatrix} -l_3 \cos(\theta 3)\ \omega 3^2 - l_4 \cos(\theta 4)\ \omega 4^2 \\ -l_3 \sin(\theta 3)\ \omega 3^2 - l_4 \sin(\theta 4)\ \omega 4^2 \end{bmatrix}$$

$$\gamma 2 := \begin{bmatrix} -l_2 \cos(\theta 2)\ \omega 2^2 \\ -l_2 \sin(\theta 2)\ \omega 2^2 \end{bmatrix}$$

$$\gamma 0 := \begin{bmatrix} -l_3 \cos(\theta 3)\ \omega 3^2 - l_4 \cos(\theta 4)\ \omega 4^2 - l_2 \cos(\theta 2)\ \omega 2^2 \\ -l_3 \sin(\theta 3)\ \omega 3^2 - l_4 \sin(\theta 4)\ \omega 4^2 - l_2 \sin(\theta 2)\ \omega 2^2 \end{bmatrix}$$

Introduzione del vettore x_2_punti

```
> v_2_punti := Vector([alpha2]);
```

$$v\_2\_punti := \begin{bmatrix} \alpha 2 \end{bmatrix}$$

```
> I_psi_u;
```

$$\left[\left[-\frac{\cos(\theta 4)}{l_3 (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}, -\frac{\sin(\theta 4)}{l_3 (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}\right], \left[\frac{\cos(\theta 3)}{l_4 (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}, \frac{\sin(\theta 3)}{l_4 (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}\right]\right] \quad \textbf{(8)}$$

calcolo del vettore y_2_punti = - Matrice_rapporti * x_2_punti - Inversa_psi_y * gamma0

```
> u_2_punti := - MatrixVectorMultiply(Rap,v_2_punti) -
  MatrixVectorMultiply(I_psi_u,gamma0);
  for ii from 1 to 2 do y_2_punti[ii] := combine(simplify
  (u_2_punti[ii]),trig) od:
  u_2_punti;
```

$$u\_2\_punti := \left[\left[-\frac{l_2 \sin(-\theta 4 + \theta 2)\ \alpha 2}{l_3 \sin(\theta 3 - \theta 4)} + \frac{\cos(\theta 4) \left(-l_3 \cos(\theta 3)\ \omega 3^2 - l_4 \cos(\theta 4)\ \omega 4^2 - l_2 \cos(\theta 2)\ \omega 2^2\right)}{l_3 (\sin(\theta 3) \cos(\theta 4) - \sin(\theta 4) \cos(\theta 3))}\right.\right.$$

$$+\frac{\sin(\theta4)\left(-l_3\sin(\theta3)\,\omega3^2-l_4\sin(\theta4)\,\omega4^2-l_2\sin(\theta2)\,\omega2^2\right)}{l_3\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}\Bigg],$$

$$\Bigg[\frac{l_2\sin(-\theta3+\theta2)\,\alpha2}{l_4\sin(\theta3-\theta4)}$$

$$-\frac{\cos(\theta3)\left(-l_3\cos(\theta3)\,\omega3^2-l_4\cos(\theta4)\,\omega4^2-l_2\cos(\theta2)\,\omega2^2\right)}{l_4\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$-\frac{\sin(\theta3)\left(-l_3\sin(\theta3)\,\omega3^2-l_4\sin(\theta4)\,\omega4^2-l_2\sin(\theta2)\,\omega2^2\right)}{l_4\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}\Bigg]\Bigg]$$

$$\Bigg[\Bigg[-\frac{l_2\sin(-\theta4+\theta2)\,\alpha2}{l_3\sin(\theta3-\theta4)}+\frac{\cos(\theta4)\left(-l_3\cos(\theta3)\,\omega3^2-l_4\cos(\theta4)\,\omega4^2-l_2\cos(\theta2)\,\omega2^2\right)}{l_3\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$+\frac{\sin(\theta4)\left(-l_3\sin(\theta3)\,\omega3^2-l_4\sin(\theta4)\,\omega4^2-l_2\sin(\theta2)\,\omega2^2\right)}{l_3\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}\Bigg],$$

$$\Bigg[\frac{l_2\sin(-\theta3+\theta2)\,\alpha2}{l_4\sin(\theta3-\theta4)}$$

$$-\frac{\cos(\theta3)\left(-l_3\cos(\theta3)\,\omega3^2-l_4\cos(\theta4)\,\omega4^2-l_2\cos(\theta2)\,\omega2^2\right)}{l_4\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$-\frac{\sin(\theta3)\left(-l_3\sin(\theta3)\,\omega3^2-l_4\sin(\theta4)\,\omega4^2-l_2\sin(\theta2)\,\omega2^2\right)}{l_4\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}\Bigg]\Bigg]$$

Confronto dei risultati ottenuti con i due metodi (la differenza delle accelerazioni trovate devono essere nulle

```
> differenza1 := u_2_punti[1]  - u_2_pu[1] ;
  differenza1 := simplify(subs(omega3=u_punto[1],differenza1)):
  differenza1 := simplify(subs(omega4=u_punto[2],differenza1)):
  differenza1 := simplify(expand(differenza1));
  differenza2 := u_2_punti[2] - u_2_pu[2];
  differenza2 := simplify(subs(omega3=u_punto[1],differenza2)):
  differenza2 := simplify(subs(omega4=u_punto[2],differenza2)):
  differenza2 := simplify(expand(differenza2));
```

$$differenza1 := -\frac{l_2\sin(-\theta4+\theta2)\,\alpha2}{l_3\sin(\theta3-\theta4)}$$

$$+\frac{\cos(\theta4)\left(-l_3\cos(\theta3)\,\omega3^2-l_4\cos(\theta4)\,\omega4^2-l_2\cos(\theta2)\,\omega2^2\right)}{l_3\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$+\frac{\sin(\theta4)\left(-l_3\sin(\theta3)\,\omega3^2-l_4\sin(\theta4)\,\omega4^2-l_2\sin(\theta2)\,\omega2^2\right)}{l_3\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$-\frac{1}{-l_3+l_3\cos(2\,\theta3-2\,\theta4)}\left(2\,l_2\,\omega4\sin(-\theta3+\theta2)\,\omega2+l_2\,\alpha2\cos(-\theta3+\theta2)\right.$$

$$-l_2\,\alpha2\cos(\theta3-2\,\theta4+\theta2)-l_2\,\omega3\,\omega2\sin(\theta3-2\,\theta4+\theta2)-l_2\,\omega3\,\omega2\sin(-\theta3+\theta2)$$

$$\left.+l_2\,\omega2^2\sin(\theta3-2\,\theta4+\theta2)-l_2\,\omega2^2\sin(-\theta3+\theta2)\right)$$

$$differenza1 := 0$$

$$differenza2 := \frac{l_2\sin(-\theta3+\theta2)\,\alpha2}{l_4\sin(\theta3-\theta4)}$$

$$-\frac{\cos(\theta3)\left(-l_3\cos(\theta3)\,\omega3^2-l_4\cos(\theta4)\,\omega4^2-l_2\cos(\theta2)\,\omega2^2\right)}{l_4\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$-\frac{\sin(\theta3)\left(-l_3\sin(\theta3)\,\omega3^2-l_4\sin(\theta4)\,\omega4^2-l_2\sin(\theta2)\,\omega2^2\right)}{l_4\left(\sin(\theta3)\cos(\theta4)-\sin(\theta4)\cos(\theta3)\right)}$$

$$-\frac{1}{-l_4+l_4\cos(2\,\theta3-2\,\theta4)}\left(-l_2\,\omega4\,\omega2\sin(-\theta4+\theta2)-l_2\,\omega4\,\omega2\sin(-2\,\theta3+\theta4\right.$$

$$+\theta2)+2\,l_2\,\omega3\sin(-\theta4+\theta2)\,\omega2-l_2\,\alpha2\cos(-2\,\theta3+\theta4+\theta2)+l_2\,\alpha2\cos(-\theta4$$

$$\left.+\theta2)-l_2\,\omega2^2\sin(-\theta4+\theta2)+l_2\,\omega2^2\sin(-2\,\theta3+\theta4+\theta2)\right)$$

$$differenza2 := 0$$

Ridefinizione del vettore v_2_punti

```
> v_2_pu;
```

$$\left[\ \alpha2\ \right]$$

VALORI NUMERICI

```
> l[1] := 80;
```

```
l[2] := 20;
l[3] := 50;
l[4] := 70;
theta2 := evalf(20*(Pi/180));
omega2 := evalf(400*(2*Pi/60));
alpha2 := 0;
```

$$l_1 := 80$$

$$l_2 := 20$$

$$l_3 := 50$$

$$l_4 := 70$$

$$\theta 2 := 0.3490658504$$

$$\omega 2 := 41.88790204$$

$$\alpha 2 := 0$$

ANALISI DELLA CONFIGURAZIONE
CALCOLO NUMERICO DELLE POSIZIONI

Espressioni da usare per l'applicazione del metodo di Newton Raphson

```
> theta3_0 := evalf(70*(Pi/180));
theta4_0 := evalf(-50*(Pi/180));
u_0 := Vector([theta3_0,theta4_0]);
I_psi_num := subs(theta3 = theta3_0,theta4 = theta4_0,I_psi_u):
ps_num := subs(theta3 = theta3_0,theta4 = theta4_0,ps):
u_1 := u_0 - MatrixVectorMultiply(I_psi_num,ps_num);
I_psi_num := subs(theta3 = u_1[1],theta4 = u_1[2],I_psi_u):
ps_num :=    subs(theta3 = u_1[1],theta4 = u_1[2],ps):
u_2 := u_1 - MatrixVectorMultiply(I_psi_num,ps_num);
I_psi_num := subs(theta3 = u_2[1],theta4 = u_2[2],I_psi_u):
ps_num :=    subs(theta3 = u_2[1],theta4 = u_2[2],ps):
u_3 := u_2 - MatrixVectorMultiply(I_psi_num,ps_num);
I_psi_num := subs(theta3 = u_3[1],theta4 = u_3[2],I_psi_u):
ps_num :=    subs(theta3 = u_3[1],theta4 = u_3[2],ps):
u_4 := u_3 - MatrixVectorMultiply(I_psi_num,ps_num);
```

$$theta3\_0 := 1.221730477$$

$$theta4\_0 := -0.8726646262$$

$$u\_0 := \begin{bmatrix} 1.221730477 \\ -0.8726646262 \end{bmatrix}$$

$$u\_1 := \begin{bmatrix} 1.23136977982799989 \\ -0.880815834225999961 \end{bmatrix}$$

$$u\_2 := \begin{bmatrix} 1.23134314725993988 \\ -0.880796908817589985 \end{bmatrix}$$

$$u\_3 := \begin{bmatrix} 1.23134314758847086 \\ -0.880796908715914206 \end{bmatrix}$$

$$u\_4 := \begin{bmatrix} 1.23134314681429880 \\ -0.880796908651079846 \end{bmatrix}$$

Angoli in gradi delle posizioni della biella e del bilanciere

```
> evalf(u_4[1]*180/Pi);
  evalf(u_4[2]*180/Pi);
```

$$70.55076545$$

$$-50.46594547$$

VALORI CALCOLATI

verifica della bontà della soluzione numerica sulle posizioni

valore del vettore dei resti

```
> scartoX := evalf(subs(theta3= u_4[1],theta4=u_4[2],psi[1]));
  scartoY := evalf(subs(theta3= u_4[1],theta4=u_4[2],psi[2]));
```

$$scartoX := 1.\ 10^{-8}$$

$$scartoY := -1.\ 10^{-8}$$

GRAFICO DEL QUADRILATERO

```
> with(geometry):
  rag_cer := 1;
  quadri := {};
  Ax := 0;
  Ay := 0;
  Bx := evalf(l[2]*cos(theta2));
  By := evalf(l[2]*sin(theta2));
```

```
Cx := Bx + evalf(subs(theta3 = u_4[1],l[3]*cos(theta3)));
Cy := By + evalf(subs(theta3 = u_4[1],l[3]*sin(theta3)));
Dx := l[1];
Dy := 0;
point(AP,Ax,Ay):
point(BP,Bx,By):
point(CP,Cx,Cy):
point(DP,Dx,Dy):

circle(APC,[AP,rag_cer]):
circle(BPC,[BP,rag_cer]):
circle(CPC,[CP,rag_cer]):
circle(DPC,[DP,rag_cer]):

segment(corp1,[AP,DP]):
segment(corp2,[AP,BP]):
segment(corp3,[BP,CP]):
segment(corp4,[CP,DP]):

quadri := quadri union {APC(color='cyan',thickness='3',filled=
'true'),corp1(color='orange',thickness='2')}:
quadri := quadri union {BPC(color='cyan',thickness='3',filled=
'true'),corp2(color='orange',thickness='2')}:
quadri := quadri union {CPC(color='cyan',thickness='3',filled=
'true'),corp3(color='orange',thickness='2')}:
quadri := quadri union {DPC(color='cyan',thickness='3',filled=
'true'),corp4(color='orange',thickness='2')}:

draw(quadri,axes='framed');
```

$$rag\_cer := 1$$

$$quadri := \{\ \}$$

$$Ax := 0$$

$$Ay := 0$$

$$Bx := 18.79385242$$

$$By := 6.840402866$$

$$Cx := 35.44242867$$

$$Cy := 53.98724697$$

$$Dx := 80$$

$$Dy := 0$$



Grafico delle velocità

```
> distance(AP,BP);
  distance(BP,CP);
  distance(CP,DP);
  evalf(distance(DP,AP));
  midpoint(G2,corp2);
  midpoint(G3,corp3);
  midpoint(G4,corp4);
  detail(CP);
  detail(G2);
  u_punto[2];
  u_pu := evalf(subs(theta3=u_4[1],theta4=u_4[2],u_punto));
```

20.00000000

50.00000000

69.99999999

$$80.00000000$$

$G2$

$G3$

$G4$

name of the object	*CP*
form of the object	*point2d*
coordinates of the point	[35.44242867, 53.98724697]

name of the object	*G2*
form of the object	*point2d*
coordinates of the point	[9.396926210, 3.420201433]

$$-\frac{11.96797201 \sin(\theta 3 - 0.3490658504)}{\sin(\theta 3 - \theta 4)}$$

$$u\_pu := \begin{bmatrix} -18.42528425 \\ -10.78335924 \end{bmatrix}$$

```
> PG2 := Vector([coordinates(G2)[1],coordinates(G2)[2]]);
Omega[2] := Matrix(2,2,[[0,-omega2],[omega2,0]]);
VG2 := MatrixVectorMultiply(Omega[2],PG2);
Omega[3] := Matrix(2,2,[[0,-u_pu[1]],[u_pu[1],0]]);
PB := 2*PG2;
BG3x := evalf((l[3]/2)*cos(u_4[1]));
BG3y := evalf((l[3]/2)*sin(u_4[1]));
BG3 := Vector([BG3x,BG3y]);
PG3 := PB + BG3;
VG3_B := MatrixVectorMultiply(Omega[3],BG3);
VB := 2 * VG2;
VG3 := VG3_B + VB;
VC := 2 * VG3_B + VB;
PC := PB + 2 * BG3;
CG4x := evalf((l[4]/2)*cos(u_4[2]));
CG4y := evalf((l[4]/2)*sin(u_4[2]));
CG4 := Vector([CG4x,CG4y]);
PG4 := PC + CG4;
Omega[4] := Matrix(2,2,[[0,-u_pu[2]],[u_pu[2],0]]);
VG4_C := MatrixVectorMultiply(Omega[4],CG4);
VG4 := VG4_C + VC;
```

$$PG2 := \begin{bmatrix} 9.396926210 \\ 3.420201433 \end{bmatrix}$$

$$\Omega_2 := \begin{bmatrix} 0 & -41.88790204 \\ 41.88790204 & 0 \end{bmatrix}$$

$$VG2 := \begin{bmatrix} -143.265062582571630 \\ 393.617524561588482 \end{bmatrix}$$

$$\Omega_3 := \begin{bmatrix} 0 & 18.42528425 \\ -18.42528425 & 0 \end{bmatrix}$$

$$PB := \begin{bmatrix} 18.7938524200000004 \\ 6.84040286599999980 \end{bmatrix}$$

$$BG3x := 8.324288125$$

$$BG3y := 23.57342205$$

$$BG3 := \begin{bmatrix} 8.324288125 \\ 23.57342205 \end{bmatrix}$$

$$PG3 := \begin{bmatrix} 27.1181405450000028 \\ 30.4138249160000030 \end{bmatrix}$$

$$VG3\_B := \begin{bmatrix} 434.347002016467741 \\ -153.377374882024554 \end{bmatrix}$$

$$VB := \begin{bmatrix} -286.530125165143260 \\ 787.235049123176964 \end{bmatrix}$$

$$VG3 := \begin{bmatrix} 147.816876851324480 \\ 633.857674241152381 \end{bmatrix}$$

$$VC := \begin{bmatrix} 582.163878867792164 \\ 480.480299359127855 \end{bmatrix}$$

$$PC := \begin{bmatrix} 35.4424286699999982 \\ 53.9872469660000008 \end{bmatrix}$$

$$CG4x := 22.27878567$$

$$CG4y := -26.99362349$$

$$CG4 := \begin{bmatrix} 22.27878567 \\ -26.99362349 \end{bmatrix}$$

$$PG4 := \begin{bmatrix} 57.7212143400000032 \\ 26.9936234759999998 \end{bmatrix}$$

$$\Omega_4 := \begin{bmatrix} 0 & 10.78335924 \\ -10.78335924 & 0 \end{bmatrix}$$

$$VG4\_C := \begin{bmatrix} -291.081939281972552 \\ -240.240149310574083 \end{bmatrix}$$

$$VG4 := \begin{bmatrix} 291.081939585819612 \\ 240.240150048553772 \end{bmatrix}$$

```
> scala_V := 0.02;
ext_V2_x := PG2[1] + scala_V * VG2[1];
ext_V2_y := PG2[2] + scala_V * VG2[2];
point(G2P,PG2[1],PG2[2]):
point(EV2P,ext_V2_x,ext_V2_y):
segment(V2,[G2P,EV2P]):
ext_V3_x := PG3[1] + scala_V * VG3[1];
ext_V3_y := PG3[2] + scala_V * VG3[2];
point(G3P,PG3[1],PG3[2]):
point(EV3P,ext_V3_x,ext_V3_y):
segment(V3,[G3P,EV3P]):
ext_V4_x := PG4[1] + scala_V * VG4[1];
ext_V4_y := PG4[2] + scala_V * VG4[2];
point(G4P,PG4[1],PG4[2]):
point(EV4P,ext_V4_x,ext_V4_y):
segment(V4,[G4P,EV4P]):
```

$$scala\_V := 0.02$$

$$ext\_V2\_x := 6.531624958$$

$$ext\_V2\_y := 11.29255192$$

$$ext\_V3\_x := 30.07447809$$

$$ext\_V3\_y := 43.09097840$$

$$ext\_V4\_x := 63.54285313$$

$$ext\_V4\_y := 31.79842648$$

```
> quadri := quadri union {V2(color='green',thickness='2')}:
quadri := quadri union {V3(color='green',thickness='2')}:
quadri := quadri union {V4(color='green',thickness='2')}:
```

```
draw(quadri);
```




```
> simplify(u_2_punti[1]);
  simplify(u_2_pu[1]);
```

$$-\frac{1}{\sin(\theta 3)\cos(\theta 4) - 1.\sin(\theta 4)\cos(\theta 3)}\Big(2.000000000\,10^{-7}\big(5.000000\,10^{6}\cos(\theta 4)\cos(\theta 3)\,\omega 3^{2} + 3.297562461\,10^{9}\cos(\theta 4) + 5.000000\,10^{6}\sin(\theta 4)\sin(\theta 3)\,\omega 3^{2} + 7.000000\,10^{6}\,\omega 4^{2} + 1.200214581\,10^{9}\sin(\theta 4)\big)\Big)$$

$$\frac{1}{-1. + \cos(2.\,\theta 3 - 2.\,\theta 4)}\Big(2.000000000\,10^{-8}\big(-1.675516082\,10^{9}\,\omega 4\sin(\theta 3 - 0.3490658504) - 8.37758041\,10^{8}\,\omega 3\sin(\theta 3 - 2.\,\theta 4 + 0.3490658504) + 8.37758041\,10^{8}\,\omega 3\sin(\theta 3 - 0.3490658504) + 3.509192674\,10^{10}\sin(\theta 3 - 2.\,\theta 4 + 0.3490658504) + 3.509192674\,10^{10}\sin(\theta 3 - 0.3490658504)\big)\Big)$$

grafico metodo dello psi_y_punto

```
> u_2_p := u_2_pu:
```

calcolo delle accelerazioni e definizione delle matrici accelerazione

```
> A3 := eval(u_2_p[1],[theta3 = u_4[1],theta4 = u_4[2],omega3=u_pu
  [1],omega4=u_pu[2]]);
  A4 := eval(u_2_p[2],[theta3 = u_4[1],theta4 = u_4[2],omega3=u_pu
  [1],omega4=u_pu[2]]);
  Alpha2 := Matrix(2,2,[[0,-alpha2],[alpha2,0]]);
  Alpha3 := Matrix(2,2,[[0,-A3],[A3,0]]);
  Alpha4 := Matrix(2,2,[[0,-A4],[A4,0]]);
```

$$A3 := -259.6554637$$

$$A4 := 584.7107572$$

$$A2 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

$$A3 := \begin{bmatrix} 0 & 259.6554637 \\ -259.6554637 & 0 \end{bmatrix}$$

$$A4 := \begin{bmatrix} 0 & -584.7107572 \\ 584.7107572 & 0 \end{bmatrix}$$

```
> scala_A := 0.0005;
  AG2 := MatrixVectorMultiply(Omega[2]^2,PG2) +
  MatrixVectorMultiply(Alpha2,PG2) ;
  ext_A2_x := PG2[1] + scala_A* AG2[1]:
  ext_A2_y := PG2[2] + scala_A* AG2[2]:
  point(EA2P,ext_A2_x,ext_A2_y):
  segment(AcG2,[G2P,EA2P]):
  AcB := 2 * AG2;
  AG3 := AcB + MatrixVectorMultiply(Omega[3]^2,BG3) +
  MatrixVectorMultiply(Alpha3,BG3) ;
  AcC := AcB + MatrixVectorMultiply(Omega[3]^2,(2*BG3)) +
  MatrixVectorMultiply(Alpha3,(2*BG3)) ;
  ext_A3_x := PG3[1] + scala_A* AG3[1]:
  ext_A3_y := PG3[2] + scala_A* AG3[2]:
  point(EA3P,ext_A3_x,ext_A3_y):
  segment(AcG3,[G3P,EA3P]):
```

```
DG4 := PG4 - Vector([l[1],0]);
AG4 := MatrixVectorMultiply(Omega[4]^2,DG4) +
MatrixVectorMultiply(Alpha4,DG4);
ext_A4_x := PG4[1] + scala_A* AG4[1]:
ext_A4_y := PG4[2] + scala_A* AG4[2]:
point(EA4P,ext_A4_x,ext_A4_y):
segment(AcG4,[G4P,EA4P]):
```

$$scala\_A := 0.0005$$

$$AG2 := \begin{bmatrix} -16487.8123100631128 \\ -6001.07290721322988 \end{bmatrix}$$

$$AcB := \begin{bmatrix} -32975.6246201262256 \\ -12002.1458144264598 \end{bmatrix}$$

$$AG3 := \begin{bmatrix} -29680.6785164577886 \\ -22166.5596827844820 \end{bmatrix}$$

$$AcC := \begin{bmatrix} -26385.7324127893408 \\ -32330.9735511425024 \end{bmatrix}$$

$$DG4 := \begin{bmatrix} -22.2787856599999970 \\ 26.9936234759999998 \end{bmatrix}$$

$$AG4 := \begin{bmatrix} -13192.8661894993056 \\ -16165.4867506805458 \end{bmatrix}$$

```
> quadri := quadri union {AcG2(color='red',thickness='3')}:
quadri := quadri union {AcG3(color='red',thickness='3')}:
quadri := quadri union {AcG4(color='red',thickness='3')}:
draw(quadri);
```



50
40
30
20
10
0
0
10
20
30
40
50
60
70
80